Quadriken 3d Paraboloid

Prof. Dr. Dörte Haftendorn: Mathematik mit MuPAD 4, Juni 07   Update 5.07.07
Web: http://haftendorn.uni-lueneburg.de   www.mathematik-verstehen.de
##
Rückwärts aufgestellt, siehe Datei Konstruktion-rückwärts 3d

```
p:=matrix([x,y,z]):pt:=linalg::transpose(p)
```

$\begin{pmatrix} x & y & z \end{pmatrix}$

```
EW:=[0,1,2]: m:=[3,-1,-3] //liegt drauf
Quadrik:=matrix([[5*x^2 - 10*x*y - 4*x*z - 40*x + 5*y^2
+ 4*y*z + 52*y + 8*z^2 + 64*z + 164]]);
```

$$\left(5 \cdot x^2 - x \cdot y \cdot 10 - x \cdot z \cdot 4 - x \cdot 40 + 5 \cdot y^2 + 4 \cdot y \cdot z + 52 \cdot y + 8 \cdot z^2 + 64 \cdot z + 164 \right)$$

```
Qp:=plot::Implicit3d(Quadrik[1]=0,x=-2..7,y=-5..0,z=-7..
0,FillColor=[0,1,0,1]):
plot(Qp):
```



A und a passend aufstellen

```
A:=matrix([[5,-5,-2],[-5,5,2],[-2,+2,8]]);
a:=matrix([-40,52,64]): at:=linalg::transpose(a);
d:=164;
```

$$\begin{pmatrix} 5 & -5 & -2 \\ -5 & 5 & 2 \\ -2 & 2 & 8 \end{pmatrix}$$

$\begin{pmatrix} -40 & 52 & 64 \end{pmatrix}$

164


```
expand(pt*A*p+at*p+d);  //Probe, ob man A,a und d
richtig hat
%-Quadrik
```

```
%-Quadrik
```

$$\left(5\cdot x^2 - x\cdot y\cdot 10 - x\cdot z\cdot 4 - x\cdot 40 + 5\cdot y^2 + 4\cdot y\cdot z + 52\cdot y + 8\cdot z^2 + 64\cdot z + 164\right)$$

$$(0)$$

hier muss 0 herauskommen

Hauptachsentransformation

```
E3:=matrix([[1,0,0],[0,1,0],[0,0,1]])
```

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

```
evli:=linalg::eigenvectors(A) //Probe, was MuPAD liefert
```

$$\left[\left[0, 1, \left[\begin{pmatrix} 1 \\ 1 \\ 0 \end{pmatrix}\right]\right], \left[6, 1, \left[\begin{pmatrix} \frac{1}{2} \\ -\frac{1}{2} \\ 1 \end{pmatrix}\right]\right], \left[12, 1, \left[\begin{pmatrix} -1 \\ 1 \\ 1 \end{pmatrix}\right]\right]\right]$$

Eigenwerte und Eigenvektoren

```
ew1 :=evli[1][1];  ew2 :=evli[2][1]; ew3 :=evli[3][1];
ev1:=evli[1][3][1]:
ev2:=-evli[2][3][1]://neg, damit es ein Rechtssystem
wird
ev3:=evli[3][3][1]:
```

$$0$$

$$6$$

$$12$$

```
linalg::det(ev1.ev2.ev3)
```

$$3$$

Diese Determinante sollte positiv sein, donst ist später noch eine Spiegelung im Spiel.

```
ev1n:=linalg::normalize(ev1):
ev2n:=linalg::normalize(ev2):
ev3n:=linalg::normalize(ev3):
P:=ev1n.ev2n.ev3n: Pt:=linalg::transpose(P);
```

$$\begin{pmatrix} \frac{\sqrt{2}}{2} & \frac{\sqrt{2}}{2} & 0 \\ -\frac{\sqrt{2}\cdot\sqrt{3}}{6} & \frac{\sqrt{2}\cdot\sqrt{3}}{6} & -\frac{\sqrt{2}\cdot\sqrt{3}}{3} \\ -\frac{\sqrt{3}}{3} & \frac{\sqrt{3}}{3} & \frac{\sqrt{3}}{3} \end{pmatrix}$$

Vektorschreibweise für die Abbildung $\vec{p} = P\,\vec{p}\,'$ und die

Vektorschreibweise für die Abbildung $\vec{p} = P\,\vec{p}\,'$ und die Quadrikgleichungen, die sich durch Einsetzen ergeben:

$$Q: \quad \vec{p}^{\,T} A\,\vec{p} + \vec{a}^{\,T}\vec{p} + d = 0$$

$$Q': \quad \vec{p}\,'^{T} P^{T} A\,P\,\vec{p}\,' + \vec{a}^{\,T} P\vec{p}\,' + d = 0$$

$$Q': \quad \vec{p}\,'^{T} D_{EW}\,\vec{p}\,' + \vec{a}^{\,T} P\vec{p}\,' + d = 0$$

mit $D_{EW} = \begin{pmatrix} \lambda_1 & 0 \\ 0 & \lambda_2 \end{pmatrix}$

bzw. in 3D

```
quastrich:=Simplify(pt*Pt*A*P*p+at*P*p+d)
```

$$\left(6\cdot y^2 - \sqrt{6}\cdot y\cdot 6 + 12\cdot z^2 + 52\cdot\sqrt{3}\cdot z + 6\cdot\sqrt{2}\cdot x + 164\right)$$

```
quastrichp:=plot::Implicit3d(quastrich[1],x=-5..4,y=-5..
6, z=-7..0,FillColor=[1,1,0,1]):
plot(quastrichp , Scaling=Constrained):
```




```
quastrich
```

$$\left(6\cdot y^2 - \sqrt{6}\cdot y\cdot 6 + 12\cdot z^2 + 52\cdot\sqrt{3}\cdot z + 6\cdot\sqrt{2}\cdot x + 164\right)$$

Die Arbeitsweise ist dieselbe wie bei der Herstellung der Scheitelform einer Parabel. Hier durch Hinsehen:

```
yterm:=hold(6*(y-1/2*sqrt(6))^2);expand(yterm);
zterm:=hold(12*(z+13/6*sqrt(3))^2);expand(zterm);
```


$$6\cdot\left(y - \frac{\sqrt{6}}{2}\right)^2$$

$$6 \cdot \left(y - \frac{\sqrt{6}}{2}\right)^2$$

$$6 \cdot y^2 - \sqrt{6} \cdot y \cdot 6 + 9$$

$$12 \cdot \left(z + \frac{13 \cdot \sqrt{3}}{6}\right)^2$$

$$12 \cdot z^2 + 52 \cdot \sqrt{3} \cdot z + 169$$

```
dd:=d-9-169;
```

$$-14$$

```
yterm+zterm+6*sqrt(2)*x+dd;
expand(%)-quastrich
```

$$6 \cdot \left(y - \frac{\sqrt{6}}{2}\right)^2 + 12 \cdot \left(z + \frac{13 \cdot \sqrt{3}}{6}\right)^2 + 6 \cdot \sqrt{2} \cdot x - 14$$

$$(0)$$

Hier muss 0 herauskommen

```
xterm:=hold(6*sqrt(2)*(x-7/6*sqrt(2)));expand(xterm);
```

$$6 \cdot \sqrt{2} \cdot \left(x - \frac{7 \cdot \sqrt{2}}{6}\right)$$

$$6 \cdot \sqrt{2} \cdot x - 14$$

Also

```
quastrichK:=xterm+yterm+zterm
```

$$6 \cdot \left(y - \frac{\sqrt{6}}{2}\right)^2 + 12 \cdot \left(z + \frac{13 \cdot \sqrt{3}}{6}\right)^2 + 6 \cdot \sqrt{2} \cdot \left(x - \frac{\sqrt{2} \cdot 7}{6}\right)$$

```
quastrich-expand(quastrichK)
```

$$(0)$$

hier muss 0 herauskommen

Letzter Teil der Hauptachsentransformation ist die Translation t

```
t:=matrix([-7/6*sqrt(2),-1/2*sqrt(6), 13/6*sqrt(3)]);
tt:=linalg::transpose(t):
tp:=plot::Arrow3d(t):
```


$$\left(\begin{array}{c} -\frac{7 \cdot \sqrt{2}}{6} \\ -\frac{\sqrt{6}}{\ldots} \end{array}\right.$$

$$\begin{pmatrix} -\frac{7\cdot\sqrt{2}}{6} \\ -\frac{\sqrt{6}}{2} \\ \frac{13\cdot\sqrt{3}}{6} \end{pmatrix}$$

$\vec{p}\,'' = \vec{p}\,' + \vec{t}$ also $\vec{p}\,' = \vec{p}\,'' - \vec{t}$ Das ergibt:

```
quaH:=Simplify(expand((pt-tt)*Pt*A*P*(p-t)+at*P*(p-t)+d)
);
```

$$\left(6\cdot y^2 + 12\cdot z^2 + 6\cdot\sqrt{2}\cdot x\right)$$

Angabe der Gleichung in der üblichen Form:

```
hold(x=-y^2/sqrt(2)-sqrt(2)*z^2)
```

$$x = -\frac{y^2}{\sqrt{2}} - \sqrt{2}\cdot z^2$$

```
quadrikHp:=plot::Implicit3d(quaH[1],x=-4..0,y=-3..3,
z=-2..2,
Scaling=Constrained):
plot(%)
```



Bestimmung des ursprünglichen Mittelpunktes:

$$\vec{m}\,'' = \begin{pmatrix} 0 \\ 0 \end{pmatrix}, \quad \vec{m}\,' = -\vec{t}, \quad \vec{m} = P(-\vec{t})$$

nun in 3D


```
m:=Simplify(P*(-t))
```

$$\begin{pmatrix} \frac{17}{6} \\ -\frac{1}{2} \\ \dots \end{pmatrix}$$

$$\begin{pmatrix} \frac{17}{6} \\ -\frac{1}{2} \\ -\frac{19}{6} \end{pmatrix}$$

Dieses ist der alte Scheitelpunkt.

Bestimmung des Urbildes des rechten Hauptscheitels:

$$\vec{r}\,'' = \begin{pmatrix} r \\ 0 \end{pmatrix}, \vec{r}\,' = \begin{pmatrix} r \\ 0 \end{pmatrix} - \vec{t}, \vec{r} = P\begin{pmatrix} r \\ 0 \end{pmatrix} + \vec{m},\ r = r \cdot \overrightarrow{ev_1} + \vec{m}$$

Mit ev1 rechts ist hier der normierte 1. Eigenvektor gemeint. Alles jetzt in 3D

```
r:=-ev1: // Achse des Paraboloids
rs:=Pt*r;
rp:=plot::Arrow3d(m,m+5*r, LineColor=[0,0,0],
LineWidth=1 ):
rsp:=plot::Arrow3d(-t, -t+5*rs, LineColor=[0,0,0],
LineWidth=1 ):
rssp:=plot::Arrow3d(5*rs, LineColor=[0,0,0],
LineWidth=1):
```

$$\begin{pmatrix} -\sqrt{2} \\ 0 \\ 0 \end{pmatrix}$$

```
pkt:=matrix([3,-1,-3]):
pktp:=plot::Point3d(pkt,PointColor=[1,0,1],PointSize=2,PointStyle=Filled
Squares):
pkts:=Pt*pkt; float(%);
pktsp:=plot::Point3d(pkts,PointColor=[1,0,1],
PointSize=2,PointStyle=FilledSquares):
pktss:=pkts+t;float(%);
pktssp:=plot::Point3d(pktss,PointColor=[1,0,1],
PointSize=2,PointStyle=FilledSquares):
```

$$\begin{pmatrix} \sqrt{2} \\ \frac{\sqrt{2}\cdot\sqrt{3}}{3} \\ -\frac{7\cdot\sqrt{3}}{3} \end{pmatrix}$$

$$\begin{pmatrix} 1.414213562 \\ 0.8164965809 \\ -4.041451884 \end{pmatrix}$$

$$\begin{pmatrix} -\frac{\sqrt{2}}{6} \\ \frac{\sqrt{2}\cdot\sqrt{3}}{} - \frac{\sqrt{6}}{} \end{pmatrix}$$

$$\begin{pmatrix} -\frac{\sqrt{2}}{6} \\ \frac{\sqrt{2}\cdot\sqrt{3}}{3}-\frac{\sqrt{6}}{2} \\ -\frac{\sqrt{3}}{6} \end{pmatrix}$$

$$\begin{pmatrix} -0.2357022604 \\ -0.4082482905 \\ -0.2886751346 \end{pmatrix}$$

```
mp:=plot::Point3d(m,PointSize=2, PointColor=[0,1,0]):
msp:=plot::Point3d(-t,PointSize=2, PointColor=[0,1,0]):
Op:=plot::Point3d([0,0,0],PointSize=2, PointColor=[0,1,0]):
ev1urp:=plot::Arrow3d(m,m+3*ev1):
ev2urp:=plot::Arrow3d(m,m+3*ev2):
ev3urp:=plot::Arrow3d(m,m+3*ev3):
tp:=plot::Arrow3d(-t,[0,0,0],LineColor=[1,0,0]):
plot(Qp,quadrikHp,quastrichp,mp,msp,Op,rp,rsp,rssp,tp,pktp,pktsp,pktssp,
      ev1urp,ev2urp,ev3urp,PointSize=2,Scaling=Constrained,
Axes=Origin);
```



Test ob es eine Drehung allein oder gefolgt von Spiegelung ist. det=1 reine Drehung

```
linalg::det(Pt)
```

$$1$$

```
plot(Qp,quastrichp, quadrikHp,mp,Op,tp,rp,rp,rsp,

ev1urp,ev2urp,ev3urp,Axes=Origin,Scaling=Constrained)
```






##
Weitere Untersuchungen, in anderer Datei